

EXAMPLE 12-1: IMPLEMENTING A REAL-TIME CLOCK USING A TIMER1 INTERRUPT SERVICE

```
RTCinit
    MOVLW    80h          ; Preload TMR1 register pair
    MOVWF    TMR1H        ; for 1 second overflow
    CLRF     TMR1L
    MOVLW    b'00001111'  ; Configure for external clock,
    MOVWF    T1CON        ; Asynchronous operation, external oscillator
    CLRF     secs         ; Initialize timekeeping registers
    CLRF     mins
    MOVLW    .12
    MOVWF    hours
    BSF      PIE1, TMR1IE ; Enable Timer1 interrupt
    RETURN

RTCisr
    BSF      TMR1H, 7      ; Preload for 1 sec overflow
    BCF      PIR1, TMR1IF ; Clear interrupt flag
    INCF     secs, F       ; Increment seconds
    MOVLW    .59
    CPFSGT   secs         ; 60 seconds elapsed?
    RETURN                    ; No, done
    CLRF     secs         ; Clear seconds
    INCF     mins, F       ; Increment minutes
    MOVLW    .59
    CPFSGT   mins         ; 60 minutes elapsed?
    RETURN                    ; No, done
    CLRF     mins         ; clear minutes
    INCF     hours, F      ; Increment hours
    MOVLW    .23
    CPFSGT   hours        ; 24 hours elapsed?
    RETURN                    ; No, done
    CLRF     hours        ; Reset hours
    RETURN                    ; Done
```